
ABSTRACT  
In clinical trials and many other research fields, proc transpose are used very often.  When many variables with their 
individual formats are transposed into one variable, we lose the formats. We can do a series of if then statements to 
put the formats back in.  However, when the variables involved are too many, the above method can be very tedious. 
This paper illustrates how to extract formats from dictionary.columns or sashelp.vcolumn, and then use PUTN 
function to assign the formats at run time and make the task much easier.  In addition, it is much easier to apply the 
same method to other projects without using a lot of if then statements in the SAS® program.  Efficiency is largely 
increased with this method. 

 
INTRODUCTION 
Problems arise when transposing the different observations with different formats into one variable.  It is impossible to 
have different formats for different observations for the same variable.  Therefore the original formats do not show up 
at all after transpose.  I have the following formats before transpose. 
 

PROC FROMAT; 

   VALUE sexcode 

     1 = 'Male  '   

     2 = 'Female'   

     ; 

   VALUE racecode 

     1 = 'White'   

     2 = 'Black'   

     3 = 'Hispanic'   

     4 = 'Asian'   

     5 = 'Other'   

     6 = 'Multi-racial'   

     ; 

………… 

RUN; 

 

 

DATA demo2; 

SET desk.demo;  

  FORMAT   

  datestmp datetime17. 

  birthdt mmddyy10. 

  sexcd sexcode. 

         racecd racecode. 

…………… 

; 

RUN; 

Following is the snapshot before the transpose.  Please note that the sexcd and racecd variables display text formats. 

 

Display1. Snapshot Before Transpose 

 
Following is the snapshot after the transpose. Please note the sexcd and racecd now are numeric, instead of text. 

 

Display 2. Snapshot After Transpose 

IF THEN AND ELSE METHOD 
There is some work needed to fix the problem.  One way of doing it is to have a series of if then statements to 
accommodate value labels and hard code the value labels into a new variable.  We can do it individually to each 
variable with format.  Following is the code that I used with this method. 

 
IF UPCASE(COMPRESS(_name_))="DATESTMP" THEN DO; 

 n_value=PUT(Col1, datetime17.); 

END; 

 

ELSE IF UPCASE(COMPRESS(_name_))="BIRTHDT" THEN DO; 

 n_value=PUT(Col1,mmddyy10. ); 

END; 

 

ELSE IF UPCASE(COMPRESS(_name_))="SEXCD" THEN DO;  

 IF Col1=1 THEN n_value= 'Male  ' ;  

      ELSE IF Col1=2 THEN n_value= 'Female' ; 

END; 

 

ELSE IF UPCASE(COMPRESS(_name_))="RACECD" THEN DO;  

IF Col1=1 THEN n_value= 'White      ' ;  

ELSE IF Col1=2 THEN n_value= 'Black'   ; 

ELSE IF Col1=3 THEN n_value= 'Hispanic'  ;  

ELSE IF Col1=4 THEN n_value= 'Asian'   ; 

ELSE IF Col1=5 THEN n_value= 'Other' ;  

ELSE IF Col1=6 THEN n_value= 'Multi-racial'  ; 

END; 

 

ELSE n_value=Col1; 

........ 

 
Following is the snapshot after I used if, then and else method. 

 

Display 3. Snapshot Using If Then Statements 

While the method is working well with only a few variables with formats to be transposed, it becomes very tedious 
when the observations are in the hundreds or thousands.  

 
EXTRACTING FORMATS FROM DICTIONARY.COLUMNS OR SASHELP.VCOLUMN AND 
APPLYNG PUTN FUNCTION 
An easier way of doing it is to pull out the format information from dictionary. columns using sql, and then apply putn 
function to assign the individual format for each observation at run time.  If you are not a sql person, you can do the 
same by pulling out format information from sashelp.vcolumn.  Following is the demonstration using 
dictionary.columns or sashelp.vcolumn then applying the formats using putn function.  This method is much easier to 
adapt to other situations without a lot of hard coding.  
 
Following is where we can find the sashelp.vcolumn file. 

 

Display 4. Where to Find Sashelp.Vcolumn File  
 

Following is how sashelp.vcolumn file looks like. 

Display 5. How Sashelp.Vcolumn File Looks Like 

 
You can use the following code to print the variable names and formats from sashelp.vcolumn.    

PROC PRINT DATA=sashelp.vcolumn NOOBS; 

WHERE libname ='WORK' and memname='SAMPLE'; 

VAR name format; 

RUN; 

Following is the snapshot of the output window after running the above code. 

 

Display 6. To See All the Formats in Sashelp.Vcolumn File Using the Above Code 

You can also export the sashelp.vcolumn file to excel if you are more familiar with Excel. 

 

Display7. How To View Sashelp.Vcolumn File in Excel 

 
Following is the snapshot of the Excel file with all other data from Sashelp.Vcolumn except the sample file (that we 
are interested) filtered out. 

 

Display8. Sashelp.Vcolumn File in Excel Formats 

Following is the code using proc sql and then print the output file to Excel using ods tagsets.excelxp. 
 

PROC TRANSPOSE DATA=sample NAME=name OUT=t89 ; 

RUN; 

 

DATA Z89; 

SET t89; 

seq+1; 

RUN; 

 

 
/*PULL OUT INFORMATION FROM DICTIONARY.COLUMNS*/ 

PROC SQL NOPRINT; 

CREATE TABLE temp AS SELECT name, format FROM dictionary.columns  

 WHERE libname='WORK' and memname="SAMPLE" ; 

QUIT; 

 

/*PULL OUT INFORMATION FROM SASHELP.VCOLUMN*/ 

/*IT DOSE THE SAME AS THE ABOVE METHOD*/ 

/*USE THIS ONE OR THE ABOVE SQL METHOD*/ 

DATA temp2; 

SET sashelp.vcolumn; 

WHERE libname ='WORK' and memname='SAMPLE'; 

KEEP name format; 

RUN; 

 

PROC SORT DATA=temp NODUPKEY; 

BY name; 

RUN; 

 

PROC SORT DATA=Z89; 

BY name; 

RUN; 

 

DATA f89; 

LENGTH name $32.; 

MERGE Z89(in=a) temp; 

BY name; 

IF a; 

IF format gt ' ' THEN value=PUTN(Col1,format); 

IF value=' ' THEN value=col1; 

RUN; 

 

PROC SORT DATA=f89; 

BY seq; 

RUN; 

 

ODS LISTING CLOSE; 

ODS tagsets.excelxp FILE="desktop\data_for_visual_check.xls"; 

ODS tagsets.excelxp OPTIONS(SHEET_NAME="id 89"); 

PROC PRINT DATA=f89 NOOBS; 

WHERE _label_ ^=' '; 

VAR _label_/STYLE=[CELLWIDTH=750]; 

VAR value /STYLE=[CELLWIDTH=300]; 

TITLE1; 

RUN; 

 

ODS tagsets.excelxp CLOSE; 

ODS LISTING; 

Following is the snapshot of the file after applying formats using PUTN function. 

 

Display 9. Snapshot of The File After Applying Formats Using PUTN Function 

 
CONCLUSION 
The latter method certainly makes your life  a lot easier.  Not only it has minimized typing when the variables involved 
are too many, but is also more adaptable to other projects.  It is always beneficial to write the programs that are 
easier to adapt to new situations, even though at first it takes more time to develop, in the long run, it really saves 
time.   

